

OPEN FILE-881



J.A.M. Hunter
Resource Geophysics Subdivision
Seismic Methods Section

Introduction

The computer routines are written for an Apple II computer with 48K bytes memory and Epson MX100 Printer and at least one disk drive. Two disk drives are required for programs FASTNMO, FASTFILT and FASTNMOF.

Port No. 1 in the Apple II is used with the Apple II Epson Interface Kit Type 2A.

Port No. 2 in the Apple II is used with the RS-232 Interface to the Nimbus Model G724S tape recorder for input of seismic data.

Game paddles are required for interactive work with the video.

Contents

- 1) NIMTODISK
- 2) NIMTODISK 1
- 3) EPLLOT
- 4) EPLLOTVA
- 5) COMOFFSEC
- 6) SYN
- 7) SYNVA
- 8) T-X MULTIMODEL
- 9) FASTNMO
- 10) FASTFILT
- 11) FASTNMOF
- 12) P-PICKER

This document was produced
by scanning the original publication.

Ce document est le produit d'une
numérisation par balayage
de la publication originale.

NIMTODISK

```
0 REM PROGRAM TRANSFERS DATA FROM NIMBUS G724S TAPE RECORDER TO
  DISK
1 REM TAPE DATA MUST BE IN 8-BIT WORD FORMAT
2 REM TAPE OUTPUT MUST BE CONNECTED TO SERIAL INTERFACE IN PORT
  #2
3 REM SET BAUD RATE OF TAPE AND INTERFACE EQUAL AT 9600
4 REM DISK FILE IN SEQUENTIAL FORMAT
5 REM G.S.C. HEADER DATA AND 8 DIGIT NIMBUS TAPE HEADER WRITTEN
  FIRST ON TAPE
6 REM PROGRAM TRANSFERS EVERY FOURTH SAMPLE TO DISK
7 REM REVISED AUG.1982
10 REM INPUT PARAMETERS
20 DIM GN(12),X%(256),H(8),T(22)
30 D$ = CHR$(4)
50 INPUT "INPUT FILE NAME ";R$
60 INPUT "INPUT SHOT OFFSET ";SF
70 INPUT "INPUT DIRECTION ";DR$
80 INPUT "INPUT GROUP INTERVAL ";GI
90 INPUT "INPUT NO OF STACKS ";SK
100 INPUT "IF ALL TRACE GAINS ARE THE SAME PUNCH 0";Z
110 IF Z = 0 THEN INPUT "INPUT GAIN";GN(1): FOR I = 2 TO 12:GN(
  I) = GN(1): NEXT I: GOTO 130
120 IF Z < > 0 THEN PRINT "INPUT TRACE GAINS ONE AT A TIME": FOR
  I = 1 TO 12: INPUT GN(I): NEXT I
130 INPUT "INPUT HC FILTER ";HC
140 INPUT "INPUT LC FILTER ";LC
150 INPUT "INPUT TIME DELAY ";TD
160 INPUT "INPUT TIME SCALE ";TS
170 HOME : VTAB (10): FOR I = 1 TO 40: PRINT "*";: NEXT I: INVERSE
  : FLASH : PRINT
180 PRINT " CHECK YOUR TAPE HEADER NUMBER "
185 PRINT " INSERT OUTPUT DATA DISK "
190 NORMAL : FOR I = 1 TO 40: PRINT "*";: NEXT I
200 PRINT : PRINT : INPUT "YES OR NO ? ";Q$
210 IF Q$ < > "N" AND Q$ < > "Y" THEN GOTO 200
220 IF Q$ = "N" THEN PRINT "DO IT NOW": GOTO 200
230 IF Q$ = "Y" THEN HOME : GOTO 240
240 PRINT "PUSH PLAYBACK ON NIMBUS TAPE"
260 PRINT D$;"PR#2"
270 PRINT ""
280 POKE 60,0
290 POKE 61,32
300 POKE 62,61
310 POKE 63,128
320 CALL - 14019
330 PRINT D$;"PR#0"
340 REM TRANSFERS EVERY FOURTH SAMPLE TO DISK
360 PRINT "PUSH ABORT BUTTON ON NIMBUS"
370 T(0) = 0:T(1) = 1:T(2) = 2:T(3) = 3:T(4) = 4:T(5) = 5:T(6) =
  6:T(7) = 7:T(8) = 8:T(9) = 9
380 T(17) = 10:T(18) = 11:T(19) = 12:T(20) = 13:T(21) = 14:T(22) =
  15
390 FOR I = 1 TO 8
400 H(I) = PEEK (8193 + I) - 48: NEXT I
410 PRINT SF: PRINT DR$: PRINT GI: PRINT SK
420 FOR I = 1 TO 12: PRINT GN(I): NEXT I
430 PRINT HC: PRINT LC: PRINT TD: PRINT TS
```

NIMTODISK

```
440 PRINT H(1);H(2);H(3);H(4);H(5);H(6);H(7);H(8)
450 J = - 6
470 PRINT D$;"OPEN ";R$
480 PRINT D$;"WRITE ";R$
490 PRINT SF: PRINT DR$: PRINT GI: PRINT SK
500 FOR I = 1 TO 12: PRINT GN(I): NEXT I
510 PRINT HC: PRINT LC: PRINT TD: PRINT TS
520 FOR I = 1 TO 8: PRINT H(I): NEXT I
530 FOR K = 0 TO 11
540 KK = 0
550 FOR I = 1 TO 1024 STEP 4
560 KK = KK + 1
570 J = J + 7
580 A = PEEK (8205 + J) - 48:A = T(A)
590 J = J + 1
600 B = PEEK (8205 + J) - 48:B = T(B)
610 X%(KK) = 16 * A + B - 127
620 NEXT I
630 J = J + 4
640 FOR I = 1 TO 256: PRINT X%(I): NEXT I
650 NEXT K
660 PRINT D$;"CLOSE ";R$
690 HOME : VTAB (12): FOR I = 1 TO 40: PRINT "*";: NEXT I
691 D$ = ""
700 PRINT SPC( 8)"RECORD TRANSFER COMPLETE"
710 FOR I = 1 TO 40: PRINT "*";: NEXT I
720 PRINT : PRINT : PRINT "FOR DATA DISK CATALOG PRESS C "
725 PRINT : PRINT : PRINT "FOR NEXT RECORD TRANSFER PRESS RETURN
"
730 GET Q$: IF Q$ < > CHR$ (13) AND Q$ < > CHR$ (67) THEN END

740 IF Q$ = CHR$ (13) THEN GOTO 1000
750 IF Q$ = CHR$ (67) THEN GOSUB 2000
751 D$ = CHR$ (4)
755 PRINT D$;"CATALOG"
760 PRINT "FOR NEXT RECORD TRANSFER PRESS RETURN"
765 GET Q$
770 IF Q$ < > CHR$ (13) THEN END
780 GOTO 1000
1000 REM PARAMETER CHANGE ROUTINE
1110 CALL - 936
1120 PRINT SPC( 12)"PARAMETER LIST"
1130 PRINT "1-FILE NAME ": PRINT
1140 PRINT "2-SHOT OFFSET ": PRINT
1150 PRINT "3-DIRECTION ": PRINT
1160 PRINT "4-GROUP SPACING ": PRINT
1170 PRINT "5-NUMBER OF STACKS": PRINT
1180 PRINT "6-TRACE GAINS": PRINT
1190 PRINT "7-HC FILTER": PRINT
1200 PRINT "8-LC FILTER": PRINT
1210 PRINT "9-TIME DELAY": PRINT
1220 PRINT "10-TIME SCALE": PRINT
1230 PRINT "11-READY TO RUN": PRINT
1240 PRINT "WHICH PARAMETERS DO YOU WISH TO CHANGE"
1250 INVERSE : FLASH : INPUT "ENTER BY NUMBER ";C: NORMAL
1260 ON C GOTO 1270,1330,1400,1450,1520,1590,1680,1750,1820,1890
```

NIMTODISK

```
1270 HOME : PRINT "FILE NAME IS- ",R$
1280 INPUT "NEW FILE NAME ? ";R$
1320 GOTO 1110
1330 HOME : PRINT "SHOT OFFSET IS"
1340 PRINT SF
1350 INPUT "NEW SHOT OFFSET ";SF
1390 GOTO 1110
1400 HOME : PRINT "DIRECTION IS - ",DR$
1410 INPUT "NEW DIRECTION ";DR$
1440 GOTO 1110
1450 HOME : PRINT "GROUP SPACING IS"
1460 PRINT GI
1470 INPUT "NEW GROUP SPACING ";GI
1510 GOTO 1110
1520 HOME : PRINT "NUMBER OF STACKS IS"
1530 PRINT SK
1540 INPUT "NUMBER OF STACKS ";SK
1580 GOTO 1110
1590 HOME : PRINT "TRACE GAINS ARE "
1600 FOR I = 1 TO 12: PRINT GN(I): NEXT I
1610 INPUT "IF ALL TRACE GAINS ARE THE SAME PUNCH 0";Z
1620 IF Z = 0 THEN INPUT "INPUT GAIN";GN(1): FOR I = 2 TO 12:GN
(I) = GN(1): NEXT I: GOTO 1640
1630 IF Z < > 0 THEN PRINT "INPUT TRACE GAINS ONE AT A TIME": FOR
I = 1 TO 12: INPUT GN(I): NEXT I
1640 INPUT "DID YOU MAKE ANY MISTAKES ? ";Q$
1650 IF Q$ < > "Y" AND Q$ < > "N" THEN GOTO 1540
1660 IF Q$ = "Y" THEN GOTO 1610
1670 IF Q$ = "N" THEN GOTO 1110
1680 HOME : PRINT "HC FILTER IS-"
1690 PRINT HC
1700 INPUT "HC FILTER CHANGE ";HC
1740 GOTO 1110
1750 HOME : PRINT "LC FILTER IS "
1760 PRINT LC
1770 INPUT "LC FILTER CHANGE ";LC
1810 GOTO 1110
1820 HOME : PRINT "TIME DELAY IS "
1830 PRINT TD
1840 INPUT "TIME DELAY CHANGE ";TD
1880 GOTO 1110
1890 HOME : PRINT "TIME SCALE IS "
1900 PRINT TS
1910 INPUT "TIME SCALE CHANGE ";TS
1950 GOTO 1110
1960 GOTO 170
1970 END
2000 PRINT D$;"CATALOG"
2010 RETURN
```

NIMTODISK 1

```
1  REM  MODIFIED VERSION OF NIMTODISK
2  REM  TRANSFERS 256 SAMPLES OF A SPECIFIED PORTION OF A SEISMOG
   RAM
3  REM  -REQUIRES THE SAME INPUT AS NIMTODISK
4  REM  -OUTPUT ON DISK IS THE SAME AS NIMTODISK
5  REM  - AUG.1982
9  DIM H(8),T(22),GN(12),X%(256)
10 D$ = CHR$(4)
11 INPUT "INSERT OUTPUT DATA DISK IN DRIVE 1 - PUNCH 1 ";PQ
12 INPUT "INPUT FILE NAME ";R$
14 INPUT "INPUT SHOT OFFSET ";SF
15 INPUT "INPUT DIRECTION ";DR$
16 INPUT "INPUT GROUP INTERVAL ";GI
18 INPUT "INPUT NO OF STACKS ";SK
19 INPUT "IF ALL TRACE GAINS ARE THE SAME PUNCH 0";Z
20 IF Z = 0 THEN INPUT "INPUT GAIN";GN(1): FOR I = 2 TO 12:GN(I
) = GN(1): NEXT I: GOTO 24
22 IF Z < > 0 THEN PRINT "INPUT TRACE GAINS ONE AT A TIME": FOR
I = 1 TO 12: INPUT GN(I): NEXT I
24 INPUT "INPUT HC FILTER ";HC
26 INPUT "INPUT LC FILTER ";LC
28 INPUT "INPUT TIME DELAY OF RECORDED DATA ";TD
29 INPUT "INPUT TIME SCALE OF RECORDED DATA ";TS
37 PRINT
38 INPUT "INPUT TIME DELAY FROM START OF RECORDED DATA TO START
OF DESIRED PORTION ";DD
39 XD = TD + DD: PRINT
40 PRINT "TIME SCALE OF DESIRED PORTION OF RECORD SHOULD BE ";TS
/ 2;" OR ";TS / 4;" MSECS"
41 PRINT
42 INPUT "INPUT TIME SCALE ";DS
44 IF DS = TS / 2 THEN FAC = 2: GOTO 50
45 IF DS = TS / 4 THEN FAC = 4: GOTO 50
46 GOTO 39
50 PRINT "CHANGE COUNTER AND PUSH PLAYBACK ON NIMBUS"
55 PRINT D$;"PR#2"
56 PRINT ""
60 POKE 60,0
65 POKE 61,32
70 POKE 62,61
75 POKE 63,128
80 CALL -14019
85 PRINT D$;"PR#0"
100 PRINT "PUSH ABORT BUTTON ON NIMBUS"
170 T(0) = 0:T(1) = 1:T(2) = 2:T(3) = 3:T(4) = 4:T(5) = 5:T(6) =
6:T(7) = 7:T(8) = 8:T(9) = 9
```

NIMTODISK 1

```
171 T(17) = 10:T(18) = 11:T(19) = 12:T(20) = 13:T(21) = 14:T(22) =
15
180 FOR I = 1 TO 8
190 H(I) = PEEK (8193 + I) - 48: NEXT I
192 PRINT SF: PRINT DR$: PRINT GI: PRINT SK
194 FOR I = 1 TO 12: PRINT GN(I): NEXT I
195 PRINT HC: PRINT LC: PRINT XD: PRINT DS
196 PRINT H(1);H(2);H(3);H(4);H(5);H(6);H(7);H(8)
300 REM - TRANSFER DESIRED PORTION OF RECORD TO DISK
310 NS = INT (DD / (TS / 1000))
315 IF FAC = 2 THEN KK = NS * 2 - 2:JJ = 3: GOTO 320
316 IF FAC = 4 THEN KK = NS * 2:JJ = 1
320 PRINT D$;"OPEN ";R$
325 PRINT D$;"WRITE ";R$
330 PRINT SF: PRINT DR$: PRINT GI: PRINT SK
331 FOR I = 1 TO 12: PRINT GN(I): NEXT I
333 PRINT HC: PRINT LC: PRINT XD: PRINT DS
335 FOR I = 1 TO 8: PRINT H(I): NEXT I
345 J = 0
350 FOR K = 0 TO 11
355 J = J + KK
360 FOR I = 1 TO 256
370 J = J + JJ
375 A = PEEK (8205 + J) - 48:A = T(A)
380 J = J + 1
385 B = PEEK (8205 + J) - 48:B = T(B)
390 X%(I) = 16 * A + B - 127
395 NEXT I
400 J = 2052 * (K + 1)
410 FOR I = 1 TO 256: PRINT X%(I): NEXT I
420 NEXT K
430 PRINT D$;"CLOSE ";R$
450 PRINT "RECORD TRANSFER COMPLETE"
470 PRINT : PRINT "CATALOG OF DISK IN DRIVE 1": PRINT
471 PRINT D$;"CATALOG"
480 PRINT
481 INPUT "PRINT E TO END PROGRAM NOW OR C TO CONTINUE ";Q$
485 IF Q$ = "E" THEN GOTO 999
486 IF Q$ < > "C" THEN GOTO 480
490 INPUT "IS ONLY CHANGE FROM PREVIOUS RUN GOING TO BE THE OUTP
UT FILE NAME? (Y/N) ";Q$
495 IF Q$ = "Y" THEN GOTO 510
496 IF Q$ < > "N" THEN GOTO 490
500 PRINT "ENTER ALL HEADER INFO FOR NEXT RECORD ": GOTO 11
510 INPUT "INPUT FILE NAME FOR NEXT RECORD ";R$: GOTO 50
999 END
```

E P L O T

```
1  REM  PROGRAM E P L O T
3  REM  COPIES SEISMIC RECORDS IN WIGGLE TRACE MODE ON VIDEO
4  REM  COPIES VIDED ON EPSON MX-100 PRINTER
5  REM  SEISMOGRAMS MUST RESIDE ON DISK IN STANDARD FORMAT AS GIV
   EN BY NIMTODISK PROGRAM
6  REM  ALL SEISMOGRAM FILES MUST RESIDE ON SAME DISK,DRIVE 1 IN
   FORT#6
7  REM  ROUTINES WRITTEN FOR APPLEII EPSON INTERFACE KIT TYPE2A
   IN FORT#1
10 REM  - PROGRAM REVISED AUG/82
100 DIM G(12),H(8),X%(11,256)
101 DIM A$(20)
110 D$ = CHR$(4)
111 INPUT "INPUT NO OF RECORDS ";NR
114 PRINT "INPUT RECORD NAMES ONE AT A TIME"
115 FOR I = 1 TO NR
120 INPUT A$(I)
125 NEXT I
130 INPUT "ENTER GAIN ";GG
135 FOR M = 1 TO NR
140 PRINT D$;"OPEN ";A$(M)
150 PRINT D$;"READ ";A$(M)
160 INPUT SF
170 INPUT E$
180 INPUT GI
190 INPUT SK
200 FOR I = 1 TO 12: INPUT G(I): NEXT I
210 INPUT HC
220 INPUT LC
230 INPUT DL
240 INPUT TS
250 FOR I = 1 TO 8: INPUT H(I): NEXT I
260 FOR L = 0 TO 11
270 FOR I = 1 TO 256: INPUT X%(L,I): NEXT I
280 NEXT L
290 PRINT D$;"CLOSE ";A$(M)
300 PRINT D$;"PR#1"
310 PRINT : PRINT : PRINT
320 PRINT "RECORD NO ";A$(M)
330 PRINT
340 PRINT "SHOT OFFSET ";SF
350 PRINT "DIRECTION ";E$
360 PRINT "GROUP INTERVAL ";GI
370 PRINT "NO. OF STACKS ";SK
380 PRINT "CHANNEL GAINS 1 TO 12 - ";
390 FOR I = 1 TO 12: PRINT G(I); SPC(1);: NEXT I
```

E P L O T

```
400 PRINT
410 IF LC = 0 THEN PRINT "LOW CUT FILTER OUT": GOTO 430
420 PRINT "LOW CUT FILTER ";LC
430 IF HC = 999 THEN PRINT "HIGH CUT FILTER OUT": GOTO 450
440 PRINT "HIGH CUT FILTER ";HC
450 PRINT "TAPE IDENT. NO. ";
460 FOR I = 1 TO 8: PRINT H(I);: NEXT I
470 PRINT
480 ST = (TS / 2) + DL:TS = TS + DL
485 PRINT
490 PRINT DL; SPC( 39);ST; SPC( 39);TS
500 PRINT D$;"PR#0"
510 HGR2
520 HCOLOR= 3
530 SY = - 8
540 FOR L = 0 TO 11
545 SY = SY + 16
550 XP = 1:YP = SY
560 FOR I = 1 TO 256
570 YY = SY + X%(L,I) * 66
580 IF YY < 1 THEN YY = 1
590 IF YY > 191 THEN YY = 191
600 HPLOT XP,YP TO I,YY
610 XP = I:YP = YY
620 NEXT I
630 NEXT L
640 XX = - 24
650 FOR I = 1 TO 10
660 XX = XX + 25
670 HPLOT XX,1 TO XX,191
680 NEXT I
690 HPLOT 125,1 TO 125,191
700 HPLOT 250,1 TO 250,191
705 HPLOT 251,1 TO 251,191
710 PRINT D$;"PR#1"
720 POKE 1913,66
730 Q$ = CHR$(17)
740 PRINT Q$
750 PRINT D$;"PR#0"
760 TEXT
770 NEXT M
780 IF Z$ = "Y" THEN GOTO 120
800 END
```


EPLOTVA

```
1  REM   PROGRAM EPLOTVA
3  REM   COPIES SEISMIC RECORDS IN VARIABLE AREA MODE ON VIDEO
4  REM   COPIES VIDEO ON EPSON MX-100 PRINTER
5  REM   SEISMOGRAMS MUST RESIDE ON DISK IN STANDARD FORMAT AS GIV
      EN BY NIMTODISK PROGRAM
6  REM   ALL SEISMOGRAM FILES MUST RESIDE ON SAME DISK,DRIVE 1 IN
      PORT#6
7  REM   ROUTINES WRITTEN FOR APPLEII EPSON INTERFACE KIT TYPE2A
      IN PORT#1
10 REM   - PROGRAM REVISED AUG/82
100 DIM G(12),H(8),X%(11,256)
101 DIM A$(20)
110 D$ = CHR$(4)
111 INPUT "INPUT NO OF RECORDS ";NR
114 PRINT "INPUT RECORD NAMES ONE AT A TIME"
115 FOR I = 1 TO NR
120 INPUT A$(I)
125 NEXT I
130 INPUT "ENTER GAIN ";GG
135 FOR M = 1 TO NR
140 PRINT D$;"OPEN ";A$(M)
150 PRINT D$;"READ ";A$(M)
160 INPUT SF
170 INPUT E$
180 INPUT GI
190 INPUT SK
200 FOR I = 1 TO 12: INPUT G(I): NEXT I
210 INPUT HC
220 INPUT LC
230 INPUT DL
240 INPUT TS
250 FOR I = 1 TO 8: INPUT H(I): NEXT I
260 FOR L = 0 TO 11
270 FOR I = 1 TO 256: INPUT X%(L,I): NEXT I
280 NEXT L
290 PRINT D$;"CLOSE ";A$(M)
300 PRINT D$;"PR#1"
310 PRINT : PRINT : PRINT
320 PRINT "RECORD NO ";A$(M)
330 PRINT
340 PRINT "SHOT OFFSET ";SF
350 PRINT "DIRECTION ";E$
360 PRINT "GROUP INTERVAL ";GI
370 PRINT "NO. OF STACKS ";SK
380 PRINT "CHANNEL GAINS 1 TO 12 - ";
390 FOR I = 1 TO 12: PRINT G(I); SPC(1);: NEXT I
400 PRINT
```

EPLOTVA

```
410 IF LC = 0 THEN PRINT "LOW CUT FILTER OUT": GOTO 430
420 PRINT "LOW CUT FILTER ";LC
430 IF HC = 999 THEN PRINT "HIGH CUT FILTER OUT": GOTO 450
440 PRINT "HIGH CUT FILTER ";HC
450 PRINT "TAPE IDENT. NO. ";
460 FOR I = 1 TO 8: PRINT H(I);: NEXT I
470 PRINT
480 ST = (TS / 2) + DL:TS = TS + DL
485 PRINT
490 PRINT DL; SPC( 39);ST; SPC( 39);TS
500 PRINT D$;"PR#0"
510 HGR2
520 HCOLOR= 3
530 SY = - 8
540 FOR L = 0 TO 11
545 SY = SY + 16 --
550 XP = 1:YP = SY
560 FOR I = 1 TO 256
570 YY = SY + X%(L,I) * GG
580 IF YY < 1 THEN YY = 1
590 IF YY > 191 THEN YY = 191
600 HPLOT XP,YP TO I,YY
605 IF X%(L,I) < 0 THEN HPLOT I,(SY - 1) TO I,YY
610 XP = I:YP = YY
620 NEXT I
630 NEXT L
640 XX = - 24
650 FOR I = 1 TO 10
660 XX = XX + 25
670 HPLOT XX,1 TO XX,191
680 NEXT I
690 HPLOT 125,1 TO 125,191
700 HPLOT 250,1 TO 250,191
705 HPLOT 251,1 TO 251,191
710 PRINT D$;"PR#1"
720 POKE 1913,66
730 Q$ = CHR$(17)
740 PRINT Q$
750 PRINT D$;"PR#0"
760 TEXT
770 NEXT M
780 IF Z$ = "Y" THEN GOTO 120
800 END
```

COMOFFSEC

```
10 REM PROGRAM COMOFFSEC*****
15 REM COPIES SEISMOGRAMS FROM DISK FILE TO VIDEO
20 REM RECORDS ARE COPIED FROM SCREEN TO AN EPSON PRINTER
25 REM RECORDS ARE PLOTTED CONTINUOSLY IN SECTION FORM
30 REM UP TO 13 RECORDS CAN BE PRINTED USING 1 INPUT
40 REM RECORDS MUST RESIDE ON 1 DISK
50 REM GAIN TAPER AND ABILITY TO USE AGC FUNCTION
55 REM SUGGESTED PARAMETERS WINDOW LENGHT-40,CONSTANT-20
60 REM ABILITY TO MUTE FRONT END OF RECORDS
70 REM REQUIRES APFLEII EPSON INTERFACE KIT IN PORT#1
80 REM INPUT DATA ON DISK IN FORMAT GIVEN BY NIMTODISK
90 REM INPUT DATA ON DISK PORT#6,DRIVE 1
100 DIM A$(14),X%(11,256)
101 DIM WT(14,14)
105 D$ = CHR$(4)
110 INPUT "ENTER TIME SCALE ";TS
111 DT = TS / 250
115 INPUT "HOW MANY RECORDS? ";N
116 PRINT "INPUT FILE NAMES ONE PER LINE"
117 FOR I = 1 TO N: INPUT A$(I): NEXT I
118 GOTO 9000
120 PRINT : PRINT
121 PRINT "PUT IN DATA DISK"
122 PRINT : PRINT
125 INPUT "INPUT INITIAL GAIN ";G
126 GA = G
130 INPUT "INPUT GAIN TAPER MULTIPLIER ";GT
131 GZ = (GT - 1) / 256
135 PRINT : INPUT "IS AGC TO BE APPLIED? (Y/N) ";AQ$
136 IF AQ$ < > "Y" AND AQ$ < > "N" THEN GOTO 135
137 IF AQ$ = "N" THEN GOTO 145
138 INPUT "INPUT AGC WINDOW LENGTH ";R2
139 R1 = INT (R2 / 2)
140 INPUT "INPUT AGC CONSTANT ";RQ
141 GQ = G * RQ * R2
142 INPUT "INPUT MUTE TIME ";MT
143 MT = MT / DT
145 PRINT D$;"PR#1"
146 PRINT : PRINT
150 PRINT "RECORDS ";A$(1);" TO ";A$(N)
151 PRINT
152 PRINT "GAIN TAPER MULTIPLIER = ";GT
153 PRINT "INITIAL GAIN = ";G
154 PRINT
155 IF AQ$ = "N" THEN GOTO 160
156 PRINT "AGC WINDOW LENGTH = ";R2
157 PRINT "AGC CONSTANT = ";RQ
160 PRINT D$;"PR#0"
161 PRINT
165 INPUT "ARE STATICS ALL ZERO? (Y/N) ";Q$
166 IF Q$ < > "Y" AND Q$ < > "N" THEN GOTO 165
167 IF Q$ = "N" THEN GOTO 170
168 FOR I = 1 TO N: FOR L = 0 TO 11:WT(I,L) = 0: NEXT L: NEXT I
169 GOTO 210
170 FOR I = 1 TO N
171 PRINT : PRINT "RECORD ";A$(I)
172 PRINT "INPUT STATICS ONE PER LINE"
173 FOR L = 0 TO 11: INPUT WT(I,L): NEXT L
```

COMOFFSEC

```
175 GOTO 9100
180 NEXT I
210 IX = 8
215 HGR2
216 HCOLOR= 3
220 FOR M = 1 TO N
230 PRINT D$;"OPEN ";A$(M)
240 PRINT D$;"READ ";A$(M)
250 INPUT A: INPUT Z$
260 FOR I = 1 TO 26: INPUT A: NEXT I
270 FOR I = 0 TO 11
280 FOR J = 1 TO 256: INPUT X%(I,J): NEXT J
290 NEXT I
300 PRINT D$;"CLOSE ";A$(M)
310 FOR I = 0 TO 11
320 SS = WT(M,I) / DT
331 IX = IX + 7
332 IF IX < 177 THEN GOTO 350
333 HPLOT 1,1 TO 1,191
334 HPLOT 125,1 TO 125,191
335 HPLOT 250,1 TO 250,191
342 IX = 15:Q$ = CHR$(17)
343 PRINT D$;"PR#1"
344 POKE 1913,66
345 PRINT Q$
346 PRINT D$;"PR#0"
347 HGR2
348 HCOLOR= 3
350 FOR J = 1 TO 256
351 GG = 1 + (GZ * J)
352 AB = X%(I,J)
360 IF AQ$ = "Y" AND J < (256 - R1) THEN GOTO 3000
365 IF MT > J THEN GA = G
366 XX = IX + AB * GA * GG
370 XX = INT (XX)
380 IF XX < 1 THEN XX = 1
385 IF XX > 191 THEN XX = 191
390 JT = J - SS
395 IF JT < 1 OR JT > 256 THEN GOTO 420
400 IF AB < 0 THEN HPLOT JT,(IX - 1) TO JT,XX: GOTO 420
410 HPLOT JT,XX
420 NEXT J
430 NEXT I
440 NEXT M
441 HPLOT 1,1 TO 1,191
442 HPLOT 125,1 TO 125,191
443 HPLOT 250,1 TO 250,191
450 PRINT D$;"PR#1"
460 POKE 1913,66
470 Q$ = CHR$(17)
480 PRINT Q$
490 PRINT D$;"PR#0"
500 TEXT
510 INPUT "ANOTHER PLOT Y/N ";Y$
520 IF Y$ = "Y" THEN GOTO 110
530 GOTO 9999
3000 IF J > (R1 + 1) THEN GOTO 3050
3005 IF J < > 1 THEN GOTO 3055
```

COMOFFSEC

```
3010 REM - SUM OVER FIRST R2 POINTS FOR J=1
3020 SM = 0
3025 FOR KI = 1 TO R2
3030 SM = SM + ABS (X%(I,KI + 1))
3035 NEXT KI
3040 GOTO 3055
3050 SM = SM - ABS (X%(I,J - R1)) + ABS (X%(I,J + R1))
3055 IF SM = 0 THEN GA = G: GOTO 365
3060 GA = G + GQ / SM
3070 GOTO 365
9000 REM - INPUT CORRECTION ROUTINE
9001 PRINT : PRINT
9010 PRINT "CHECK INPUT FILE NAMES"
9020 FOR I = 1 TO N: PRINT A$(I): NEXT I
9030 INPUT "OK? (Y/N) ";Q$
9035 IF Q$ < > "Y" AND Q$ < > "N" THEN GOTO 9030
9040 IF Q$ = "Y" THEN GOTO 120
9050 PRINT "REENTER FILE NAMES ONE PER LINE": GOTO 117
9100 PRINT : PRINT
9110 PRINT "CHECK STATICS FOR RECORD ";A$(I)
9120 PRINT
9130 FOR L = 0 TO 11: PRINT WT(I,L): NEXT L
9135 PRINT
9140 INPUT "OK? (Y/N) ";Q$
9141 PRINT
9142 IF Q$ < > "Y" AND Q$ < > "N" THEN GOTO 9140
9145 IF Q$ = "Y" THEN GOTO 180
9150 PRINT "REENTER STATICS FOR RECORD ";A$(I): GOTO 173
9999 END
```

SYN

```
2  REM  PROGRAM TO PRODUCE SYNETHETIC SEISMOGRAM ON VIDE0
4  REM  REQUIRED INPUT
6  REM  --NO. OF LAYERS IN MODEL
8  REM  --VELOCITY AND DEPTH OF EACH LAYER
10 REM  -JUG SPACING
12 REM  -SHOT OFFSET
14 REM  -RECORD LENGHT (TIME)
15 REM  -DELAY TIME TO START OF RECORD
16 REM  -OPTION OF REFRACTION OR REFLECTION MODEL
18 REM  -NO. OF RAY SEGMENTS
20 REM  -WAVE CODE
22 REM  -OPTION OF HARD COPY OUTPUT VIA PORT#1
24 REM  -OUTPUT IN WIGGLE TRACE FORMAT
39  LOMEM: 16384
40  DIM V(10),Z(10),MM(20)
41  DIM X%(11,279),A(20)
50  PRINT "INPUT NUMBER OF LAYERS"
51  A(1) = 0:A(2) = 4:A(3) = 8:A(4) = 10:A(5) = 8:A(6) = 2:A(7) =
    - 5:A(8) = - 8:A(9) = - 6:A(10) = - 2:A(11) = 0
60  INPUT N
70  FOR I = 1 TO N
80  PRINT "INPUT V AND Z FOR LAYER ";I
90  INPUT V(I),Z(I)
100 NEXT I
110 INPUT "INPUT JUG SPACING ";DX
120 INPUT "INPUT SHOT OFFSET ";SX
122 INPUT "INPUT TIME SCALE 50,100,200,500,1000 ";BK
123 AK = BK / 250
129 PRINT "INPUT START TIME IN MSECS": INPUT ST
142 INPUT "REFRAC-R REFLEX-F PRINT-P END-E ";E$
144 IF E$ = "P" THEN GOTO 4020
145 IF E$ = "R" THEN GOTO 3000
146 IF E$ = "E" THEN GOTO 5000
152 IF JI = 1 THEN GOTO 3000
158 PRINT "FOR REFLECTION INPUT NO OF RAY SEGMENTS"
159 PRINT "IN DOWN-GOING AND UP-GOING RAY"
160 INPUT M: IF M = 0 THEN GOTO 900
170 PRINT "INPUT WAVE CODE"
172 TB = 0
174 SB = 0
180 PRINT "ONE NUMBER AT A TIME"
190 FOR I = 1 TO M: INPUT MM(I)
200 NEXT I
210 X = SX - DX
220 S = 0
221 HGR : HCOLOR= 3
222 SC = 0
223 FOR I = 1 TO 11
224 SC = SC + 25
225 SR = SC
226 HPL0T SR,1 TO SR,149
227 NEXT I
230 FOR L = 0 TO 11
240 S = S + 12
250 X = X + DX
260 F = 0
270 DP = 0.5
```

SYN

```

271 P = P + DP
280 GOSUB 500
300 XC = ABS (XX - X)
310 IF XC < 1 THEN GOTO 700
320 XC = -XX - X
330 IF XC < 0 THEN GOTO 350
340 IF XC > 0 THEN GOTO 400
350 P = P + DP
360 GOSUB 500
370 GOTO 300
400 DP = DP / 2
410 P = P - DP
420 GOSUB 500
430 GOTO 300
500 XX = 0.0
510 FOR J = 1 TO M
520 K = MM(J)
525 B = 1 - V(K) * V(K) * P * P
526 IF B < 0 THEN XX = X * 2: GOTO 550
527 IF B < 0 THEN GOTO 525
530 XX = XX + Z(K) * V(K) * P / SQR (B)
540 NEXT J
550 RETURN
700 T = 0.0
710 FOR J = 1 TO M
720 K = MM(J)
730 T = T + Z(K) / (V(K) * SQR (1 - V(K) * V(K) * P * P))
735 NEXT J
736 T = T - ST
737 IF T < 0 THEN GOTO 850
740 TS = T / AK
741 TS = INT (TS)
750 IF TS > 279 THEN GOTO 765
758 FOR KK = 1 TO 11
760 I = TS - 1 + KK
762 IF I > 279 THEN GOTO 764
763 X%(L,I) = X%(L,I) + A(KK)
764 NEXT KK
765 XB = 1
766 YB = S
767 FOR I = 1 TO 279
770 Y = S - X%(L,I)
780 IF Y < 1 THEN Y = 1
790 IF Y > 159 THEN Y = 159
800 HPLOT XB,YB TO I,Y
810 YB = Y
820 XB = I
830 NEXT I
850 NEXT L
860 GOTO 142
900 SC = 0
1000 PRINT "DO YOU WANT ANOTHER PLOT"
1010 PRINT "IF SO PRINT 1"
1020 INPUT KL
1030 IF KL = 1 THEN GOTO 50
1040 GOTO 4000
3000 PRINT "INPUT REFRACTOR LAYER NO"
3010 INPUT IJ

```

SYN

```


3015 IF IJ = 0 THEN GOTO 900
3020 X = SX - DX
3030 PRINT "INPUT NO OF RAY SEGMENTS"
3040 PRINT "DOWN TO THE REFRACTOR"
3050 INPUT IK
3060 PRINT "INPUT WAVE CODE DOWN TO REFRACTOR"
3070 PRINT "ONE NUMBER AT A TIME"
3080 FOR L = 1 TO IK
3090 INPUT MM(L)
3100 NEXT L
3110 TB = 0
3120 SB = 0
3130 S = 0
3132 HGR : HCOLOR= 3
3133 SR = 0
3134 FOR I = 1 TO 11:SR = SR + 25: HPLOT SR,1 TO SR,149: NEXT I
3140 FOR L = 0 TO 11
3150 S = S + 12
3160 X = X + DX
3170 TE = X / V(IJ)
3180 TF = 0
3181 IF IJ = 1 THEN GOTO 3250
3185 SX = 0
3186 FOR J = 1 TO IK
3187 LL = MM(J):SN = V(LL) / V(IJ)
3188 AN = ATN (SN / SQR ( - SN * SN + 1))
3189 SX = SX + Z(LL) * TAN (AN)
3190 NEXT J
3191 SX = SX * 2
3192 IF X < SX THEN GOTO 3325
3193 FOR J = 1 TO IK
3200 LL = MM(J)
3210 TF = TF + 2 * Z(LL) * SQR (V(IJ) * V(IJ) - V(LL) * V(LL)) /
(V(IJ) * V(LL))
3220 NEXT J
3250 T = TE + TF
3260 T = T - ST
3270 IF T < 0 THEN GOTO 3350
3280 TS = T / AK
3285 TS = INT (TS)
3287 IF TS > 279 THEN GOTO 3325
3290 FOR KK = 1 TO 11
3300 I = TS - 1 + KK
3310 X%(L,I) = X%(L,I) + A(KK)
3320 NEXT KK
3325 XB = 1:YB = S
3330 FOR I = 1 TO 279
3335 Y = S - X%(L,I)
3336 IF Y < 1 THEN Y = 1
3337 IF Y > 159 THEN Y = 159
3338 HPLOT XB,YB TO I,Y
3340 YB = Y:XB = I
3345 NEXT I
3350 NEXT L
3360 GOTO 142
4000 INPUT "YA WANNA PRINTOUT Y/N ";YY$
4010 IF YY$ < > "Y" THEN GOTO 5000

```



SYN

```

4020 REM
4030 QQ$ = CHR$ (17)
4040 D$ = CHR$ (4)
4045 PRINT D$;"PR#1"
4046 PRINT "MODEL PARAMETERS"
4047 FOR I = 1 TO N
4048 IF I = N THEN PRINT "LAYER ";I,"V= ";V(N): GOTO 4050
4049 PRINT "LAYER ";I,"V= ";V(I),"Z= ";Z(I)
4050 NEXT I
4051 PRINT "JUG SPACING ";DX
4052 PRINT "SHOT OFFSET ";SX
4053 PRINT "START TIME ";ST;" MSECS"
4054 BK = BK / 10
4055 PRINT "TIMING LINES ";BK;" MS INTERVALS"
4056 POKE 1913,65
4060 PRINT QQ$
4070 PRINT D$;"PR#0"
5000 END
    
```



V(5) EXAMPLE REFRACTION TOP OF V(5)
ENTER RAY PATH (DOWN GOING) 1-2-3-4



V(5) EXAMPLE REFLECTION TOP OF V(5)
ENTER RAY PATH 1-2-3-4-4-3-2-1

SYNVA

```
2  REM PROGRAM TO PRODUCE SYNETHETIC SEISMOGRAM ON VIDEO
4  REM REQUIRED INPUT
6  REM --NO. OF LAYERS IN MODEL
8  REM --VELOCITY AND DEPTH OF EACH LAYER
10 REM -JUG SPACING
12 REM -SHOT OFFSET
14 REM -RECORD LENGHT (TIME)
15 REM -DELAY TIME TO START OF RECORD
16 REM -OPTION OF REFRACTION OR REFLECTION MODEL
18 REM -NO. OF RAY SEGMENTS
20 REM -WAVE CODE
22 REM OUTPUT VARIABLE AREA FORMAT
24 REM OPTION OF HARD COPY
39 LOMEM: 16384
40 DIM V(10),Z(10),MM(20)
41 DIM X%(11,279),A(20)
50 PRINT "INPUT NUMBER OF LAYERS"
51 A(1) = 0:A(2) = 4:A(3) = 8:A(4) = 10:A(5) = 8:A(6) = 2:A(7) =
   - 5:A(8) = - 8:A(9) = - 6:A(10) = - 2:A(11) = 0
60 INPUT N
70 FOR I = 1 TO N
80 PRINT "INPUT V AND Z FOR LAYER ";I
90 INPUT V(I),Z(I)
100 NEXT I
110 INPUT "INPUT JUG SPACING ";DX
120 INPUT "INPUT SHOT OFFSET ";SX
122 INPUT "INPUT TIME SCALE 50,100,200,500,1000 ";BK
123 AK = BK / 250
129 PRINT "INPUT START TIME IN MSECS": INPUT ST
142 INPUT "REFRAC-R REFLEX-F PRINT-P END-E ";E$
144 IF E$ = "P" THEN GOTO 4020
145 IF E$ = "R" THEN GOTO 3000
146 IF E$ = "E" THEN GOTO 5000
152 IF JI = 1 THEN GOTO 3000
158 PRINT "FOR REFLECTION INPUT NO OF RAY SEGMENTS"
159 PRINT "IN DOWN-GOING AND UP-GOING RAY"
160 INPUT M: IF M = 0 THEN GOTO 900
170 PRINT "INPUT WAVE CODE"
172 TB = 0
174 SB = 0
180 PRINT "ONE NUMBER AT A TIME"
190 FOR I = 1 TO M: INPUT MM(I)
200 NEXT I
210 X = SX - DX
220 S = 0
221 HGR : HCOLOR= 3
222 SC = 0
223 FOR I = 1 TO 11
224 SC = SC + 25
225 SR = SC
226 HPLOT SR,1 TO SR,149
227 NEXT I
228 XX = 0
230 FOR L = 0 TO 11
240 S = S + 12
250 X = X + DX
260 F = 0
270 DF = 0.5
```

SYNVA

```
271 P = P + DP
280 GOSUB 500
300 XC = ABS (XX - X)
310 IF XC < (0.01 * X) THEN GOTO 700
320 XC = XX - X
330 IF XC < 0 THEN GOTO 350
340 IF XC > 0 THEN GOTO 400
350 P = P + DP
360 GOSUB 500
370 GOTO 300
400 DP = DP / 2
410 P = P - DP
420 GOSUB 500
430 GOTO 300
500 XX = 0.0
510 FOR J = 1 TO M
520 K = MM(J)
525 B = 1 - V(K) * V(K) * P * P
526 IF B < 0 OR B = 0 THEN XX = X * 2: GOTO 550
527 IF B < 0 THEN GOTO 525
530 XX = XX + Z(K) * V(K) * P / SQR (B)
540 NEXT J
550 RETURN
700 T = 0.0
710 FOR J = 1 TO M
720 K = MM(J)
730 T = T + Z(K) / (V(K) * SQR (1 - V(K) * V(K) * P * P))
735 NEXT J
736 T = T - ST
737 IF T < 0 THEN GOTO 850
740 TS = T / AK
741 TS = INT (TS)
750 IF TS > 279 THEN GOTO 765
758 FOR KK = 1 TO 11
760 I = TS - 1 + KK
762 IF I > 279 THEN GOTO 764
763 X%(L,I) = X%(L,I) + A(KK)
764 NEXT KK
765 XB = 1
766 YB = S
767 FOR I = 1 TO 279
770 Y = S - X%(L,I)
780 IF Y < 1 THEN Y = 1
790 IF Y > 159 THEN Y = 159
800 HPLOT XB,YB TO I,Y
805 IF X%(L,I) > 0 THEN HPLOT I,(S - 1) TO I,Y
810 YB = Y
820 XB = I
830 NEXT I
850 NEXT L
860 GOTO 142
900 SC = 0
1000 PRINT "DO YOU WANT ANOTHER PLOT"
1010 PRINT "IF SO PRINT 1"
1020 INPUT KL
1030 IF KL = 1 THEN GOTO 50
1040 GOTO 4000
3000 PRINT "INPUT REFRACTOR LAYER NO"
```

SYNVA

```

3010 INPUT IJ
3015 IF IJ = 0 THEN GOTO 900
3020 X = SX - DX
3030 PRINT "INPUT NO OF RAY SEGMENTS"
3040 PRINT "DOWN TO THE REFRACTOR"
3050 INPUT IK
3060 PRINT "INPUT WAVE CODE DOWN TO REFRACTOR"
3070 PRINT "ONE NUMBER AT A TIME"
3080 FOR L = 1 TO IK
3090 INPUT MM(L)
3100 NEXT L
3110 TB = 0
3120 SB = 0
3130 S = 0
3132 HGR : HCOLOR= 3
3133 SR = 0
3134 FOR I = 1 TO 11:SR = SR + 25: HPLOT SR,1 TO SR,149: NEXT I
3140 FOR L = 0 TO 11
3150 S = S + 12
3160 X = X + DX
3170 TE = X / V(IJ)
3180 TF = 0
3181 IF IJ = 1 THEN GOTO 3250
3185 QX = 0
3186 FOR J = 1 TO IK
3187 LL = MM(J):SN = V(LL) / V(IJ)
3188 AN = ATN (SN / SQR (- SN * SN + 1))
3189 QX = QX + Z(LL) * TAN (AN)
3190 NEXT J
3191 QX = QX * 2
3192 IF X < QX THEN GOTO 3325
3193 FOR J = 1 TO IK
3200 LL = MM(J)
3210 TF = TF + 2 * Z(LL) * SQR (V(IJ) * V(IJ) - V(LL) * V(LL)) /
(V(IJ) * V(LL))
3220 NEXT J
3250 T = TE + TF
3260 T = T - ST
3270 IF T < 0 THEN GOTO 3350
3280 TS = T / AK
3285 TS = INT (TS)
3287 IF TS > 279 THEN GOTO 3325
3290 FOR KK = 1 TO 11
3300 I = TS - 1 + KK
3310 X%(L,I) = X%(L,I) + A(KK)
3320 NEXT KK
3325 XB = 1:YB = S
3330 FOR I = 1 TO 279
3335 Y = S - X%(L,I)
3336 IF Y < 1 THEN Y = 1
3337 IF Y > 159 THEN Y = 159
3338 HPLOT XB,YB TO I,Y
3339 IF X%(L,I) > 0 THEN HPLOT I,(S - 1) TO I,Y
3340 YB = Y:XB = I
3345 NEXT I
3350 NEXT L
3360 GOTO 142

```

SYNVA

```
4000 INPUT "YA WANNA PRINTOUT Y/N ";YY$
4010 IF YY$ < > "Y" THEN GOTO 5000
4020 REM
4030 QQ$ = CHR$ (17)
4040 D$ = CHR$ (4)
4045 PRINT D$;"PR#1"
4046 PRINT "MODEL PARAMETERS"
4047 FOR I = 1 TO N
4048 IF I = N THEN PRINT "LAYER ";I,"V= ";V(N): GOTO 4050
4049 PRINT "LAYER ";I,"V= ";V(I),"Z= ";Z(I)
4050 NEXT I
4051 PRINT "JUG SPACING ";DX
4052 PRINT "SHOT OFFSET ";SX
4053 PRINT "START TIME ";ST;" MSECS"
4054 BK = BK / 10
4055 PRINT "TIMING LINES ";BK;" MS INTERVALS"
4056 POKE 1913,65
4060 PRINT QQ$
4070 PRINT D$;"PR#0"
5000 END
```

T-X MULTIMODEL

```
2 REM .PROGRAM TO PRODUCE A.T-X MODEL ON VIDEO .
4 REM REQUIRED INPUT
6 REM --NO. OF LAYERS IN MODEL
8 REM --VELOCITY AND DEPTH OF EACH LAYER
10 REM -JUG SPACING -24 CHANNELS
12 REM -SHOT OFFSET
14 REM -DIGITAL RATE (DEFINES RECORD TIME
16 REM -OPTION OF REFRACTION OR REFLECTION MODEL
18 REM -NO. OF RAY SEGMENTS
20 REM -WAVE CODE
22 REM OPTION OF HARD COPY OUTPUT VIA PORT#1
40 DIM V(10),Z(10),MM(20)
50 PRINT "INPUT NUMBER OF LAYERS"
60 INPUT N
70 FOR I = 1 TO N
80 PRINT "INPUT V AND Z FOR LAYER ";I
90 INPUT V(I),Z(I)
100 NEXT I
110 PRINT "INPUT JUG SPACING": INPUT DX
120 PRINT "INPUT SHOT OFFSET": INPUT SX
122 PRINT "INPUT DIGITAL RATE IN MSECS"
124 INPUT AK
126 BK = AK / 2
128 PRINT "MULTIPLY TIME SCALE BY ";BK
129 PRINT "INPUT START TIME IN MSECS": INPUT ST
130 HGR : HCOLOR= 3
131 SC = 0
132 FOR I = 1 TO 55
133 SC = SC + 5
134 FOR J = 1 TO 149 STEP 2: HPLOT SC,J: NEXT J
135 NEXT I
```

T-X MULTIMODEL

```

136 SC = 0
137 FOR I = 1 TO 11
138 SC = SC + 25
139 SR = SC
140 H PLOT SR,1 TO SR,149
141 NEXT I
142 S = 0
143 FOR I = 1 TO 24
144 S = S + 6
145 H PLOT 1,S TO 275,S
146 NEXT I
147 GOSUB 1100
148 SS = ST + 550 * BK: PRINT "TIME WINDOW ";ST;" TO ";SS;" MSECS"
150 PRINT "REFRACTION PUNCH 1, REFLECTION PUNCH 0"
151 INPUT JI
152 IF JI = 1 THEN GOTO 3000
158 PRINT "FOR REFLECTION INPUT NO OF RAY SEGMENTS"
159 PRINT "IN DOWN-GOING AND UP-GOING RAY"
160 INPUT M: IF M = 0 THEN GOTO 900
170 PRINT "INPUT WAVE CODE"
172 TB = 0
174 SB = 0
180 PRINT "ONE NUMBER AT A TIME"
190 FOR I = 1 TO M: INPUT MM(I)
200 NEXT I
210 X = SX - DX
220 S = 0
230 FOR I = 1 TO 24
240 S = S + 6
250 X = X + DX
260 P = 0
270 DP = 0.5
271 P = P + DP
280 GOSUB 500
300 XC = ABS (XX - X)
310 IF XC < 1 THEN GOTO .700
320 XC = XX - X
330 IF XC < 0 THEN GOTO 350
340 IF XC > 0 THEN GOTO 400
350 P = P + DP
360 GOSUB 500
370 GOTO 300
400 DP = DP / 2
410 P = P - DP
420 GOSUB 500
430 GOTO 300
500 XX = 0.0
510 FOR J = 1 TO M
520 K = MM(J)
525 B = 1 - V(K) * V(K) * P * P
526 IF B < 0 THEN XX = X * 2: GOTO 550
527 IF B < 0 THEN GOTO 525
530 XX = XX + Z(K) * V(K) * P / SQR (B)
540 NEXT J
550 RETURN
700 T = 0.0
710 FOR J = 1 TO M
720 K = MM(J)
730 T = T + Z(K) / (V(K) * SQR (1 - V(K) * V(K) * P * P))

```

```
735 NEXT J
736 T = T - ST
737 IF T < 0 THEN GOTO 850
740 TS = T / AK
750 IF TS > 279 THEN GOTO 860
760 IF TB = 0 THEN TB = TS
770 IF SB = 0 THEN SB = S
780 HPLOT TB,SB TO TS,S
790 TB = TS
800 SB = S
850 NEXT I
860 GOTO 150
900 SC = 0
1000 PRINT "DO YOU WANT ANOTHER PLOT"
1010 PRINT "IF SO PRINT 1"
1020 INPUT KL
1030 IF KL = 1 THEN GOTO 50
1040 GOTO 4000
1100 HPLOT 24,150 TO 22,150
1110 HPLOT 22,150 TO 22,152
1120 HPLOT 22,152 TO 24,152
1130 HPLOT 24,152 TO 24,154
1140 HPLOT 24,154 TO 22,154
1150 HPLOT 26,150 TO 28,150
1160 HPLOT 28,150 TO 28,154
1170 HPLOT 28,154 TO 26,154
1180 HPLOT 26,154 TO 26,150
1190 HPLOT 49,150 TO 51,150
1200 HPLOT 51,150 TO 51,154
1210 HPLOT 51,154 TO 49,154
1220 HPLOT 49,154 TO 49,150
1230 HPLOT 47,150 TO 47,154
1240 HPLOT 53,150 TO 55,150
1250 HPLOT 55,150 TO 55,154
1260 HPLOT 55,154 TO 53,154
1270 HPLOT 53,154 TO 53,150
1280 HPLOT 95,150 TO 97,150
1290 HPLOT 97,150 TO 97,152
1300 HPLOT 97,152 TO 95,152
1310 HPLOT 95,152 TO 95,154
1320 HPLOT 95,154 TO 97,154
1330 HPLOT 99,150 TO 101,150
1340 HPLOT 101,150 TO 101,154
1350 HPLOT 101,154 TO 99,154
1360 HPLOT 99,154 TO 99,150
1370 HPLOT 103,150 TO 105,150
1380 HPLOT 105,150 TO 105,154
1390 HPLOT 105,154 TO 103,154
1400 HPLOT 103,154 TO 103,150
1410 HPLOT 145,150 TO 147,150
1420 HPLOT 147,150 TO 147,152
1430 HPLOT 147,152 TO 145,152
1440 HPLOT 147,152 TO 147,154
1450 HPLOT 147,154 TO 145,154
1460 HPLOT 149,150 TO 151,150
1470 HPLOT 151,150 TO 151,154
1480 HPLOT 151,154 TO 149,154
1490 HPLOT 149,154 TO 149,150
```

```
1500 HPLOT 153,150 TO 155,150
1510 HPLOT 155,150 TO 155,154
1520 HPLOT 155,154 TO 153,154
1530 HPLOT 153,154 TO 153,150
1540 HPLOT 195,150 TO 195,152
1550 HPLOT 195,152 TO 197,152
1560 HPLOT 197,151 TO 197,154
1570 HPLOT 199,150 TO 201,150
1580 HPLOT 201,150 TO 201,154
1590 HPLOT 201,154 TO 199,154
1600 HPLOT 199,154 TO 199,150
1610 HPLOT 203,150 TO 205,150
1620 HPLOT 205,150 TO 205,154
1630 HPLOT 205,154 TO 203,154
1640 HPLOT 203,154 TO 203,150
1650 HPLOT 247,150 TO 245,150
1660 HPLOT 245,150 TO 245,152
1670 HPLOT 245,152 TO 247,152
1680 HPLOT 247,152 TO 247,154
1690 HPLOT 247,154 TO 245,154
1700 HPLOT 249,150 TO 251,150
1710 HPLOT 251,150 TO 251,154
1720 HPLOT 251,154 TO 249,154
1730 HPLOT 249,154 TO 249,150
1740 HPLOT 253,150 TO 255,150
1750 HPLOT 255,150 TO 255,154
1760 HPLOT 255,154 TO 253,154
1770 HPLOT 253,154 TO 253,150
1780 HPLOT 277,4 TO 277,8
1790 HPLOT 279,34 TO 277,34
1800 HPLOT 277,34 TO 277,38
1810 HPLOT 277,38 TO 279,38
1820 HPLOT 279,38 TO 279,36
1830 HPLOT 279,36 TO 277,36
1840 HPLOT 274,70 TO 274,74
1850 HPLOT 277,70 TO 279,70
1860 HPLOT 279,70 TO 279,72
1870 HPLOT 279,72 TO 277,72
1880 HPLOT 277,72 TO 277,74
1890 HPLOT 277,74 TO 279,74
1900 HPLOT 274,106 TO 274,110
1910 HPLOT 277,106 TO 279,106
1920 HPLOT 279,106 TO 279,110
1930 HPLOT 279,110 TO 277,110
1940 HPLOT 277,110 TO 277,106
1950 HPLOT 277,108 TO 279,108
1960 HPLOT 272,142 TO 274,142
1970 HPLOT 274,142 TO 274,144
1980 HPLOT 274,144 TO 272,144
1990 HPLOT 272,144 TO 272,146
2000 HPLOT 272,146 TO 274,146
2010 HPLOT 277,142 TO 277,144
2020 HPLOT 277,144 TO 279,144
2030 HPLOT 279,142 TO 279,146
2040 RETURN
3000 PRINT "INPUT REFRACTOR LAYER NO"
3010 INPUT IJ
```



```
3015 IF IJ = 0 THEN GOTO 900
3020 X = SX - DX
3030 PRINT "INPUT NO OF RAY SEGMENTS"
3040 PRINT "DOWN TO THE REFRACTOR"
3050 INPUT IK
3060 PRINT "INPUT WAVE CODE DOWN TO REFRACTOR"
3070 PRINT "ONE NUMBER AT A TIME"
3080 FOR L = 1 TO IK
3090 INPUT MM(L)
3100 NEXT L
3110 TB = 0
3120 SB = 0
3130 S = 0
3140 FOR I = 1 TO 24
3150 S = S + 6
3160 X = X + DX
3170 TE = X / V(IJ)
3180 TF = 0
3185 IF IJ = 1 THEN GOTO 3250
3190 FOR J = 1 TO IK
3200 LL = MM(J)
3210 TF = TF + 2 * Z(LL) * SQR (V(IJ) * V(IJ) - V(LL) * V(LL)) / (V(IJ) * V(LL))
3220 NEXT J
3250 T = TE + TF
3260 T = T - ST
3270 IF T < 0 THEN GOTO 3350
3280 TS = T / AK
3290 IF TS > 279 THEN GOTO 150
3300 IF TB = 0 THEN TB = TS
3310 IF SB = 0 THEN SB = S
3320 H PLOT TB,SB TO TS,S
3330 TB = TS
3340 SB = S
3350 NEXT I
3360 GOTO 150
4000 INPUT "YA WANNA PRINTOUT Y/N ";YY$
4010 IF YY$ < > "Y" THEN GOTO 5000
4020 POKE 1913,65
4030 QQ$ = CHR$(17)
4040 D$ = CHR$(4)
4045 PRINT D$;"PR#1"
4046 PRINT "MODEL PARAMETERS"
4047 FOR I = 1 TO N
4048 IF I = N THEN PRINT "LAYER ";I,"V= ";V(N): GOTO 4050
4049 PRINT "LAYER ";I,"V= ";V(I),"Z= ";Z(I)
4050 NEXT I
4051 PRINT "JUG SPACING ";DX
4052 PRINT "SHOT OFFSET ";SX
4053 PRINT "START TIME ";ST;" MSECS"
4054 PRINT "MULTIPLY TIME SCALE BY ";BK
4060 PRINT QQ$
4070 PRINT D$;"PR#0"
5000 END
```

FASTNMO

```

1  REM   PROGRAM PERFORMS NORMAL MOVE-OUT CORRECTIONS
2  REM   NMO VELOCITIES ARE ENTERED AS VEL-TIME PAIRS
3  REM   ENTIRE TIME SECTION MUST BE DESCRIBED
4  REM   FOR EXAMPLE
5  REM   ....RECORD DIGITIZED AT 0.8 SAMPLE RATE
6  REM   ....RECORD CONTAINS 256 SAMPLES PER TRACE
7  REM   ....RECORD TOTAL TIME PERIOD IS 204.8 MSEC
8  REM   ....RECORD HAS A 20 MSEC DELAY TIME
10 REM   IN THE EXAMPLE LAST NMO VELOCITY IS THE VELOCITY AT 224
    .8 MSEC
15 REM   INPUT MUST BE IN STANDARD NIMTODISK FORMAT
20 REM   UP TO 13 SEISMOGRAPH FILES MAY BE ENTERED AT ONE TIME
25 REM   ALL INPUT MUST BE ON THE SAME DISK DRIVE#1 OF PORT#6
26 REM   ALL RECORDS MUST HAVE THE IDENTICAL SPREAD GEOMETRY
27 REM   ALL RECORDS MUST HAVE THE IDENTICAL DELAY TIMES
30 REM   OUTPUT IS ON DRIVE#2 OF PORT#6
90 REM   NORMAL MOVE-OUT ROUTINE
95 PRINT "INPUT DATA ON DISK 1, OUTPUT ON DISK 2"
100 DIM T%(12,300),H(8),TR(20),V(20),VV(257)
101 DIM F(50),XS%(12,256),A$(13),B$(13),PZ%(12,256)
102 DIM QC(18)
110 D$ = CHR$(4)
111 INPUT "INPUT DIGITAL RATE ";DT
114 INPUT "INPUT NO OF RECORDS ";KQ
116 FOR J = 1 TO KQ
120 INPUT "ENTER INPUT RECORD FILE NAME ";A$(J)
125 INPUT "ENTER OUTPUT RECORD FILE NAME ";B$(J)
126 NEXT J
150 INPUT "INPUT SHOT OFFSET IN M ";SX
160 INPUT "INPUT GROUP SPACING IN M ";DX
161 INPUT "INPUT SHIFT DELAY";LD
162 INPUT "INPUT TIME DELAY";TD
163 TD = TD / DT:TD = INT (TD)
165 SL = LD / DT:SL = INT (SL)
170 PRINT "INPUT TIMES AND NMO VEL PAIRS"
180 PRINT "FOR THE BOTTOM OF EACH ZONE"
190 PRINT "PUNCH 0,0 TO END"
195 F(1) = - 0.0275:F(2) = - 0.0735:F(3) = - 0.0458:F(4) = 0.0
    740:F(5) = 0.16
200 IK = 0
210 IK = IK + 1
220 INPUT TR(IK),V(IK)
230 IF TR(IK) = 0 THEN GOTO 250
240 GOTO 210
250 IK = IK - 1
251 IP = 0:V(0) = V(1)
252 FOR L = 1 TO IK
253 K = TR(L) / DT:K = INT (K) - TD + 1:BT = K - IP:VD = (V(L) -
    V(L - 1)) / BT:KM = 0
254 FOR I = IP TO K
255 KM = KM + 1:VV(I) = V(L - 1) + KM * VD
256 NEXT I
257 IP = K
258 NEXT L
259 X = SX - DX
260 FOR L = 1 TO 12
261 X = X + DX:XX = X * X
262 FOR I = 1 TO 256

```

FASTNMO

```

263 HH = (I + TD) * DT * VV(I) / 2
264 TT = SQR (XX + 4 * HH * HH) / VV(I)
265 TT = (TT / DT) - TD: II = INT (TT)
266 IF II > 256 THEN II = 256
267 PZ%(L,I) = II
268 NEXT I
269 NEXT L
270 FOR JH = 1 TO KQ
271 PRINT D$;"OPEN ";A$(JH);",D1"
272 PRINT D$;"READ ";A$(JH)
273 INPUT QA
274 INPUT QB$
275 FOR I = 1 TO 18: INPUT QC(I): NEXT I
280 FOR I = 1 TO 8: INPUT H(I): NEXT I
290 AZ = SL + 255
310 FOR K = 1 TO 12
320 FOR I = SL TO AZ: INPUT T%(K,I): NEXT I
350 NEXT K
360 PRINT D$;"CLOSE ";A$(JH)
364 PRINT A$(JH)
400 PRINT "STARTING NMO"
410 FOR L = 1 TO 12
420 FOR I = 1 TO 256
430 II = PZ%(L,I)
440 T%(L,I + SL) = T%(L,II + SL)
450 NEXT I
460 NEXT L
470 PRINT "END NMO"
690 PRINT "STARTING WRITE TO DISK"
710 PRINT D$;"OPEN ";B$(JH);",D2"
720 PRINT D$;"DELETE ";B$(JH);",D2"
730 PRINT D$;"OPEN ";B$(JH);",D2"
740 PRINT D$;"WRITE ";B$(JH)
741 PRINT QA
742 PRINT QB$
743 QC(17) = QC(17) - LD
745 FOR I = 1 TO 18: PRINT QC(I): NEXT I
750 FOR I = 1 TO 8
755 PRINT H(I)
760 NEXT I
762 FOR L = 1 TO 12
765 FOR I = 1 TO 256
775 PRINT T%(L,I)
780 NEXT I
830 NEXT L
840 PRINT D$;"CLOSE ";B$(JH)
860 PRINT "OUTPUT FILE NAME=" ;B$(JH)
870 NEXT JH
880 GOTO 1100
1100 END

```

FASTFILT

27

```

5  REM THIS PROGRAM CONVOLVES A BAND-PASS FILTER FUNCTION WITH EACH SEISMIC TRACE
10 REM FILTER IS SINC FUNCTION WITH A FEJER WEIGHTED WINDOW
15 REM FOR 200 MSEC RECORDS 200-600 HZ
20 REM FOR 100 MSEC RECORDS 100-300 HZ
25 REM FOR 50 MSEC RECORDS 300-1200 HZ
30 REM UP TO 13 SEISMOGRAPH FILES MAY BE INPUT AT ONE TIME
35 REM ALL SEISMOGRAPHS MUST BE ON THE SAME DISK
40 REM INPUT MUST BE IN THE FORMAT OF NIMTODISK PROGRAM
45 REM REQUIRES TWO DISK DRIVES
50 REM INPUT PORT#6,DRIVE 1
55 REM OUTPUT PORT#6,DRIVE 2
90 REM NORMAL MOVE-OUT AND FILTER ROUTINE
95 PRINT "INPUT DATA ON DISK 1, OUTPUT ON DISK 2"
100 DIM T%(12,257),H(8),TR(20),V(20),VV(257)
101 DIM F(50),XS%(12,256),A$(13),B$(13),PZ%(12,256)
102 DIM QC(18)
110 D$ = CHR$(4)
114 INPUT "INPUT NO OF RECORDS ";KQ
116 FOR J = 1 TO KQ
120 INPUT "ENTER INPUT RECORD FILE NAME ";A$(J)
125 INPUT "ENTER OUTPUT RECORD FILE NAME ";B$(J)
126 NEXT J
195 F(1) = 0:F(2) = 0.00323:F(3) = 0.00115:F(4) = - 0.00366:F(5) = 0.00258
196 F(6) = 0.00578:F(7) = - 0.0275:F(8) = - 0.0735:F(9) = - 0.0458:F(10) = 0.0
    0:F(11) = 0.16
197 FOR I = 1 TO 11:F(I) = F(I) * 1000: NEXT I
270 FOR JH = 1 TO KQ
271 PRINT D$;"OPEN ";A$(JH);",D1"
272 PRINT D$;"READ ";A$(JH)
273 INPUT QA
274 INPUT QB$
275 FOR I = 1 TO 18: INPUT QC(I): NEXT I
280 FOR I = 1 TO 8: INPUT H(I): NEXT I
310 FOR K = 1 TO 12
320 FOR I = 1 TO 256: INPUT T%(K,I): NEXT I
350 NEXT K
360 PRINT D$;"CLOSE ";A$(JH)
364 PRINT A$(JH)
570 PRINT "STARTING FILTER"
580 AM = 0
600 FOR I = 11 TO 246
610 FOR L = 1 TO 12
619 XS%(L,I) = 0
620 FOR KK = 1 TO 9
621 XS%(L,I) = XS%(L,I) + (T%(L,I - 10 + KK) + T%(L,I + 10 - KK)) * F(KK + 1)
622 NEXT KK
623 XS%(L,I) = XS%(L,I) + T%(L,I) * F(11)
624 XS%(L,I) = XS%(L,I) * 0.00336
630 NEXT L
640 NEXT I
690 PRINT "STARTING WRITE TO DISK"
710 PRINT D$;"OPEN ";B$(JH);",D2"
720 PRINT D$;"DELETE ";B$(JH);",D2"
730 PRINT D$;"OPEN ";B$(JH);",D2"
740 PRINT D$;"WRITE ";B$(JH)

```

FASTFILT

28

```
741 PRINT QA
742 PRINT QB$
743 FOR I = 1 TO 18: PRINT QC(I): NEXT I
750 FOR I = 1 TO 8
755 PRINT H(I)
760 NEXT I
762 FOR L = 1 TO 12
765 FOR I = 1 TO 256
775 PRINT XS%(L,I)
780 NEXT I
830 NEXT L
840 PRINT D$;"CLOSE ";B$(JH)
860 PRINT "OUTPUT FILE NAME= ";B$(JH)
870 NEXT JH
880 GOTO 1100
1100 END
```

FASTNMOF

```
5 REM PROGRAM PERFORMS NORMAL MOVE-OUT AND BAND PASS FILTERING
10 REM NMO VELOCITIES ARE ENTERED AS VEL-TIME PAIRS
15 REM ENTIRE TIME SECTION MUST BE DESCRIBED
20 REM FOR EXAMPLE
25 REM ....RECORD DIGITIZED AT 0.8 SAMPLE RATE
30 REM ....RECORD CONTAINS 256 SAMPLES PER TRACE
32 REM ....RECORD HAS 20 MSEC. DELAY
35 REM ....RECORD TOTAL TIME PERIOD IS 224.8 MSEC
40 REM IN THE EXAMPLE LAST NMO VELOCITY IS THE VELOCITY AT 224
    .8 MSEC.
45 REM INPUT MUST BE IN STANDARD NIMTODISK FORMAT
50 REM UP TO 13 SEISMOGRAPH FILES MAY BE INPUTTED AT ONE TIME
55 REM ALL INPUT MUST BE ON THE SAME DISK
60 REM OUTPUT IS ON DRIVE#2 OF PORT#6
65 REM FILTER IS A BAND-PASS FILTER FUNCTION IN THE TIME DOMAIN

70 REM ALL RECORDS RUN MUST HAVE THE IDENTICAL SPREAD GEOMETRY
75 REM ALL RECORDS RUN MUST HAVE THE IDENTICAL DELAY TIMES
80 REM SEE REM IN FASTFILT FOR FILTER PARAMETERS
90 REM NORMAL MOVE-OUT AND FILTER ROUTINE
95 PRINT "INPUT DATA ON DISK 1, OUTPUT ON DISK 2"
100 DIM T%(12,300),H(8),TR(20),V(20),VV(257)
101 DIM F(50),XS%(12,256),A$(13),B$(13),PZ%(12,256)
102 DIM QC(18)
110 D$ = CHR$(4)
111 INPUT "INPUT DIGITAL RATE ";DT
114 INPUT "INPUT NO OF RECORDS ";KQ
116 FOR J = 1 TO KQ
120 INPUT "ENTER INPUT RECORD FILE NAME ";A$(J)
125 INPUT "ENTER OUTPUT RECORD FILE NAME ";B$(J)
126 NEXT J
150 INPUT "INPUT SHOT OFFSET IN M ";SX
160 INPUT "INPUT GROUP SPACING IN M ";DX
161 INPUT "ENTER SHIFT DELAY";LD
162 INPUT "INPUT TIME DELAY";TD
```

FASTFILT

```

163 TD = TD / DT:TD = INT (TD)
165 SL = LD / DT:SL = INT (SL)
170 PRINT "INPUT TIMES AND NMO VEL PAIRS"
180 PRINT "FOR THE BOTTOM OF EACH ZONE"
190 PRINT "PUNCH 0,0 TO END"
195 F(1) = 0:F(2) = 0.00323:F(3) = 0.00115:F(4) = - 0.00366:F(5)
    = 0.00258
196 F(6) = 0.00578:F(7) = - 0.0275:F(8) = - 0.0735:F(9) = - 0.
    0458:F(10) = 0.0740:F(11) = 0.16
197 FOR I = 1 TO 11:F(I) = F(I) * 1000: NEXT I
200 IK = 0
210 IK = IK + 1
220 INPUT TR(IK),V(IK)
230 IF TR(IK) = 0 THEN GOTO 250
240 GOTO 210
250 IK = IK - 1
251 IP = 0:V(0) = V(1)
252 FOR L = 1 TO IK
253 K = TR(L) / DT:K = INT (K) - TD + 1:BT = K - IP:VD = (V(L) -
    V(L - 1)) / BT:KM = 0
254 FOR I = IP TO K
255 KM = KM + 1:VV(I) = V(L - 1) + KM * VD
256 NEXT I
257 IP = K
258 NEXT L
259 X = SX - DX
260 FOR L = 1 TO 12
261 X = X + DX:XX = X * X
262 FOR I = 1 TO 256
263 HH = (I + TD) * DT * VV(I) / 2
264 TT = SQR (XX + 4 * HH * HH) / VV(I)
265 TT = (TT / DT) - TD:II = INT (TT)
266 IF II > 256 THEN II = 256
267 PZ%(L,I) = II
268 NEXT I
269 NEXT L
270 FOR JH = 1 TO KQ
271 PRINT D$;"OPEN ";A$(JH);",D1"
272 PRINT D$;"READ ";A$(JH)
273 INPUT QA
274 INPUT QB$
275 FOR I = 1 TO 18: INPUT QC(I): NEXT I
280 FOR I = 1 TO 8: INPUT H(I): NEXT I
290 AZ = SL + 255
310 FOR K = 1 TO 12
320 FOR I = SL TO AZ: INPUT T%(K,I): NEXT I
350 NEXT K
360 PRINT D$;"CLOSE ";A$(JH)
364 PRINT A$(JH)
400 PRINT "STARTING NMO"
410 FOR L = 1 TO 12
420 FOR I = 1 TO 256
430 II = PZ%(L,I)
440 T%(L,I + SL) = T%(L,II + SL)
450 NEXT I
460 NEXT L
470 PRINT "END NMO"
570 PRINT "STARTING FILTER"

```

FASTNMOF

```

600 FOR I = 11 TO 246
610 FOR L = 1 TO 12
619 XS%(L,I) = 0
620 FOR KK = 1 TO 9
621 XS%(L,I) = XS%(L,I) + (T%(L,I - 10 + KK) + T%(L,I + 10 - KK))
      * F(KK + 1)
622 NEXT KK
623 XS%(L,I) = XS%(L,I) + T%(L,I) * F(11)
624 XS%(L,I) = XS%(L,I) * 0.00336
630 NEXT L
640 NEXT I
690 PRINT "STARTING WRITE TO DISK"
710 PRINT D$; "OPEN "; B$(JH); ", D2"
720 PRINT D$; "DELETE "; B$(JH); ", D2"
730 PRINT D$; "OPEN "; B$(JH); ", D2"
740 PRINT D$; "WRITE "; B$(JH)
741 PRINT QA
742 PRINT QB$
743 QC(17) = QC(17) - LD
745 FOR I = 1 TO 18: PRINT QC(I): NEXT I
750 FOR I = 1 TO 8
755 PRINT H(I)
760 NEXT I
762 FOR L = 1 TO 12
765 FOR I = 1 TO 256
775 PRINT XS%(L,I)
780 NEXT I
830 NEXT L
840 PRINT D$; "CLOSE "; B$(JH)
860 PRINT "OUTPUT FILE NAME= "; B$(JH)
870 NEXT JH
880 GOTO 1100
1100 END

```

P-PICKER

```
1800 REM A PROGRAM FOR SCREEN DISPLAY OF SEISMOGRAM
1810 REM DISPLAY FOR INTERACTIVE PICKING OF EVENTS
1820 REM GAME PADDLES ARE USED TO MOVE CURSOR
1830 REM OPTION OF 1ST HALF,MIDDLE HALF,OR END HALF DISPLAY
1840 REM INPUT DATA MUST BE IN FORMAT OF NIMTODISK PROGRAM
1850 REM MOVE CURSOR OM SCREEN WITH PADDLE KNOBS
1860 REM HORIZONTAL CONTROL SIDE BUTTON ENTERS TIME DATA
1870 REM ENTER UP TO 12 TRACE TIME PICKS
1880 REM IF LESS THAN 12 ENTER TRACE NUMBER >12 TO EXIT
1890 REM REFRACTION OR REFLECTION PICK OPTION
1900 REM LEAST SQUARES VELOCITIES COMPUTED
1910 REM OPTION OF TIME-DISTANCE DATA SAVE ON DISK OR PRINTER
```

```
1 REM REFLECTION PICK
2 GOTO 100
3 REM CURSOR MONITOR ROUTINE
4 FOR I = 1 TO 5
10 TX = PDL (0)
20 TY = PDL (1)
30 NEXT I
40 TY = TY * 159 / 255
50 FOR I = 1 TO 5: HPLOT TX,TY: NEXT I
51 HCOLOR= 0
52 HPLOT TX,TY
53 HCOLOR= 3
60 FOR I = 1 TO 5:CK = PEEK ( - 16286): NEXT I
70 IF CK > 127 THEN GOTO 90
80 GOTO 4
90 RETURN
100 DIM X(256),H(8),D(12),T(12),R(12),TT(12)
105 DIM G(12)
110 D$ = CHR$(4)
120 INPUT "ENTER INPUT FILE NAME ";A$
131 GN = GN / 2
140 INPUT "1ST HALF PNCH 1, 2ND PNCH 2, MIDDLE PNCH 3 ";PP
142 INPUT "ENTER GAIN ADJUST ";GN
144 GN = GN / 2
163 PRINT D$;"OPEN ";A$
170 PRINT D$;"READ ";A$
171 INPUT SX
172 INPUT DR$
173 INPUT DX
174 INPUT SK
175 FOR I = 1 TO 12: INPUT G(I): NEXT I
176 INPUT HC
177 INPUT LC
178 INPUT TD
179 INPUT TS
180 IF TS = 50 THEN DT = 0.2
181 IF TS = 100 THEN DT = 0.4
182 IF TS = 200 THEN DT = 0.8
183 IF TS = 500 THEN DT = 2.0
184 FOR I = 1 TO 8: INPUT H(I): NEXT I
190 HGR
200 HCOLOR= 3
210 XX = 1
220 FOR L = 1 TO 12
222 IF PP = 1 THEN GOTO 240
```


P-PICKER

```
224 IF PP = 3 THEN GOTO 232
226 FOR I = 1 TO 128: INPUT PQ: NEXT I
228 FOR I = 1 TO 128: INPUT X(I): NEXT I
230 GOTO 244
232 FOR I = 1 TO 64: INPUT PQ: NEXT I
234 FOR I = 1 TO 128: INPUT X(I): NEXT I
236 FOR I = 1 TO 64: INPUT PQ: NEXT I
238 GOTO 244
240 FOR I = 1 TO 128: INPUT X(I): NEXT I
242 FOR I = 1 TO 128: INPUT PQ: NEXT I
244 XX = XX + 12
250 XP = 1
260 YP = XX
269 II = 0
270 FOR I = 1 TO 256 STEP 2
275 II = II + 1
280 Y = XX + X(II) * GN
285 Y = INT (Y)
286 IF Y < 1 THEN Y = 1
287 IF Y > 159 THEN Y = 159
290 H PLOT XP,YP TO I,Y
300 XP = I
310 YP = Y
320 NEXT I
330 NEXT L
340 PRINT D$;"CLOSE ";A$
350 XP = 0
360 FOR I = 1 TO 10:XP = XP + 25: H PLOT XP,1 TO XP,159: NEXT I
370 PRINT "TO PICK REFLEXIONS ENTER TRACE USE CURSOR BUTTON"
380 J = 0
382 INPUT "ENTER DELAY TIME IN MS ";TQ
385 INPUT "REFLECTION PNCH 0 REFRACTION PNCH 1 ";KI
386 IF KI > 1 THEN GOTO 1000
390 J = J + 1
400 IF J > 12 THEN GOTO 520
410 INPUT "ENTER TRACE NO ";TR
420 IF TR > 12 THEN GOTO 520
430 R(J) = TR
440 TR = TR - 1
450 D(J) = SX + TR * DX
455 IF KI = 1 THEN GOTO 470
460 D(J) = D(J) * D(J)
470 GOSUB 3
480 TT(J) = TX / 2
482 T(J) = TX / 2
484 IF PP = 2 THEN T(J) = T(J) + 128
485 IF PP = 3 THEN T(J) = T(J) + 64
490 T(J) = T(J) * DT
495 T(J) = T(J) + TD - TQ
496 IF KI = 1 THEN GOTO 510
500 T(J) = T(J) * T(J)
510 GOTO 390
520 J = J - 1
530 REM LEAST SQRS FIT
540 MX = 0:MY = 0:X2 = 0:XY = 0
550 FOR I = 1 TO J
560 MX = MX + D(I)
570 MY = MY + T(I)
580 X2 = X2 + D(I) * D(I)
```

P-PICKER

```

590 XY = XY + D(I) * T(I)
600 NEXT I
610 V = (J * X2 - MX * MX) / (J * XY - MX * MY)
620 TC = (X2 * MY - MX * XY) / (J * X2 - MX * MX)
625 IF KI = 1 THEN GOTO 650
630 V = SQR (V)
640 TC = SQR (TC)
650 RP = R(1) * 12 + 1
660 SP = TT(1) * 2
670 FOR I = 2 TO J
680 RR = R(I) * 12 + 1
690 SS = TT(I) * 2
700 H PLOT SP,RP TO SS,RR
710 RP = RR
720 SP = SS
730 .NEXT I
740 PRINT "INT TIME= ";TC,"V= ";V
750 PRINT D$;"PR#1"
760 PRINT " "
770 PRINT A$
780 FOR I = 1 TO 8: PRINT H(I);
790 NEXT I
795 PRINT " "
796 IF KI = 1 THEN GOTO 805
800 PRINT "REFLEC INT T= ";TC,"VEL= ";V
804 GOTO 810
805 PRINT "REFRAC INT T= ";TC,"VEL= ";V
810 PRINT D$;"PR#0"
811 INPUT "TO OUTPUT T-X DATA TO PRINTER FNCH 1 ";IK
812 IF IK = 1 THEN GOTO 850
813 INPUT "TO OUTPUT T-X DATA TO DISK FNCH 1 ";IK
814 IF IK = 1 THEN GOTO 900
820 INPUT "TO CONTINUE ON THIS HALF FNCH 1 ";IK
830 IF IK < > 1 THEN GOTO 1000
840 GOTO 380
850 PRINT D$;"PR#1"
855 FOR I = 1 TO J
860 IF KI = 0 THEN D(I) = SQR (D(I));T(I) = SQR (T(I))
865 PRINT D(I),T(I)
870 NEXT I
875 PRINT D$;"PR#0"
880 GOTO 813
900 INPUT "INPUT DISK FILE NAME ";F$
910 INPUT "INSERT OUTPUT DISK FNCH 1 ";QL
920 PRINT D$;"OPEN ";F$
930 PRINT D$;"WRITE ";F$
940 PRINT J
950 FOR I = 1 TO J
960 IF KI = 0 THEN D(I) = SQR (D(I));T(I) = SQR (T(I))
970 PRINT D(I)
975 PRINT T(I)
980 NEXT I
985 PRINT D$;"CLOSE ";F$
990 PRINT "TRANSFER TO DISK COMPLETE"
995 GOTO 820
1000 INPUT "TO VIEW OTHER HALF FNCH 1 ";IK
1010 IF IK = 1 THEN GOTO 140
1020 END

```